

Gramática BizSpec

NO_TERMINALES

<Program>, <Rule>, <Test>, <StmtList>, <Given>, <Expect>, <Assign>, <DottedIdent>, <Expr>, <OrExpr>, <OrTail>, <AndExpr>, <AndTail>, <CmpExpr>, <CmpTail>, <AddExpr>, <AddTail>, <MulExpr>, <MulTail>, <Unary>, <Primary>

TERMINALES

rule, test, when, then, given, expect, and, or, not, true, false, null, undefined, IDENT, NUMBER, STRING, { }, (), ,, ,, =, ==, !=, >=, <=, >, <, +, -, *, /, NEWLINE, EOF

AXIOMA

<Program>

PRODUCCIONES

<Program> ::= <Program>_Body EOF

<Program>_Body ::= <Rule> <Program>_Body

<Program>_Body ::= <Test> <Program>_Body

<Program>_Body ::= NEWLINE <Program>_Body

<Program>_Body ::= EPSILON

<Rule> ::= rule STRING when <Expr> then IDENT = <Expr> <OptNewline>

<Test> ::= test STRING { <OptNewline> <StmtList> } <OptNewline>

<OptNewline> ::= NEWLINE

<OptNewline> ::= EPSILON

<StmtList> ::= <Stmt> <StmtList>

<StmtList> ::= EPSILON

<Stmt> ::= <Given>

<Stmt> ::= <Expect>

<Stmt> ::= NEWLINE

<Given> ::= given <Assign> <OptNewline>

<Expect> ::= expect <Expr> <OptNewline>

<Assign> ::= <DottedIdent> = <Expr>

<Assign> ::= IDENT = <Expr>

<DottedIdent> ::= IDENT <DottedTail>

<DottedTail> ::= . IDENT <DottedTail>
<DottedTail> ::= EPSILON

<Expr> ::= <OrExpr>
<OrExpr> ::= <AndExpr> <OrTail>
<OrTail> ::= or <AndExpr> <OrTail>
<OrTail> ::= EPSILON

<AndExpr> ::= <CmpExpr> <AndTail>
<AndTail> ::= and <CmpExpr> <AndTail>
<AndTail> ::= EPSILON

<CmpExpr> ::= <AddExpr> <CmpTail>
<CmpTail> ::= == <AddExpr> <CmpTail>
<CmpTail> ::= != <AddExpr> <CmpTail>
<CmpTail> ::= >= <AddExpr> <CmpTail>
<CmpTail> ::= <= <AddExpr> <CmpTail>
<CmpTail> ::= > <AddExpr> <CmpTail>
<CmpTail> ::= < <AddExpr> <CmpTail>
<CmpTail> ::= EPSILON

<AddExpr> ::= <MulExpr> <AddTail>
<AddTail> ::= + <MulExpr> <AddTail>
<AddTail> ::= - <MulExpr> <AddTail>
<AddTail> ::= EPSILON

<MulExpr> ::= <Unary> <MulTail>
<MulTail> ::= * <Unary> <MulTail>
<MulTail> ::= / <Unary> <MulTail>
<MulTail> ::= EPSILON

<Unary> ::= not <Unary>
<Unary> ::= <Primary>

<Primary> ::= NUMBER
<Primary> ::= STRING
<Primary> ::= true
<Primary> ::= false
<Primary> ::= null
<Primary> ::= undefined
<Primary> ::= <DottedIdent>
<Primary> ::= IDENT
<Primary> ::= (<Expr>)